

Cindyscript でストレンジアトラクタを描く

静岡県立磐田南高等学校

入谷 昭 (Akira Irtitani)

Iwataminami highschool

1 はじめに


高等学校の理数科や SSH(スーパー・サイエンス・ハイスクール) 指定校では, 自然科学の課題研究を行なっている. 理科の分野では新しい発見につながるような実験・観察も可能であるが, 数学となると高校生の方では独創的な研究をするのは難しい. しかし, コンピュータを用いるとさまざまな試行錯誤が可能となり, 非線形微分方程式の数値解やフラクタルといった, 高等学校の教育課程にない分野についてもアプローチが可能となる. このときに, 適切な数式処理ソフトを利用することが肝要になるが, 幾何ソフトとして出発した Cinderella が, 第 2 版となってプログラミング言語 CindyScript を備え, 代数計算も行なえるようになった. CindyScript は高校生にとっても習得が容易で, 計算結果をビジュアルに表示するのにも優れたツールである. 静岡県立磐田南高等学校理数科では, 昨年度の「自然界における螺旋」の研究に続き, Cinderella を用いたカオスのストレンジアトラクタの描画をテーマに課題研究を行なった. 本稿では, その成果を紹介しながら, 数学教育における Cinderella の有用性について論じていく.

2 Cindyscript とは

CindyScript は, シンデレラに 2.0 版から加わったプログラミング環境である. 特徴として次のようなものが挙げられる.

- インタプリタである.
書いたコードをすぐに実行して試すことができる.
- 描画面とは別にスクリプトエディタウィンドウを持つ.
コードの実行結果を別ウィンドウで確認でき, 初学者にも使いやすい.
- 関数型言語である.
関数の定義などは Mathematica や Maxima と類似している. 分岐や繰り返し処理も関数型で行なう. 手続き型言語を学んだ者なら簡単に理解でき, 初学者にもわかりやすい.
- 描画した幾何要素をコントロールできる.
これは, 他言語にはない大きなメリットである. 幾何ソフトとして出発した Cinderella では, 作図した幾何要素のコントロールが CindyScript から容易に行なえる. そのため, グラフィクスに関するコーディングがほとんど不要となり, アルゴリズムの中核となる部分に専念できる.

- スロットという概念により、スクリプトを機能別にグルーピングできる。
マウスやキーボードによるユーザー入力、内部時計の利用など、目的に応じてコードをグルーピングでき、構造的に見通しのよいプログラムが作成できる。
- 変数の取り扱いの柔軟性
変数の型は明示しなくてよい。Cinderellaが適切にメモリを割り当てる。幾何ソフトとして描いた図形要素の名前をそのまま変数として扱えることを含め、変数の取扱いに柔軟性がある。




3 課題研究での取り組み

静岡県立磐田南高等学校理数科2年生の課題研究で、CindyScriptを用いてストレンジアトラクタの描画を行い、微分方程式の係数をインタラクティブに変化させたり、項を追加したりして自分たちのストレンジアトラクタを作り出す試みを行なった。その結果次のような作品を作った。(取り組み期間は2年生10月から3年生5月)

3.1 係数をインタラクティブに変化させるシミュレーション


ローレンツの微分方程式では、係数を変化させるとさまざまな対流のモデルが生じる。ある係数の組合せによって表れるのがローレンツアトラクタである。この係数を、画面上のスライダ操作によりインタラクティブに変化させるものを作った。Mathematica(7以降)では、Manipulate関数によりスライダを使ってのパラメータ操作を行なえるようになったが、CindyScriptを用いればより柔軟なスライダを作成することができる。



スライダを用いて係数をインタラクティブに変化させる

3.2 3Dグラフィクスによるローレンツアトラクタの描画


Mathematicaには、微分方程式の解の例としてローレンツアトラクタがあり、方程式と初期条件を書くだけで3次元の図が描画できる。また、Macintoshのグラフ描画ソフトGrapherでも同様にして簡単に描画できる。Web上には3Dのアトラクタが描けるフリーソフトもある。しかし、数学教育という観点からは、出来合いのもので結果だけ得るのではなく、非線形微分方程式の数値解を求める原理や3Dグラフィクスの基礎も学ばせたい。短期間の課題研究では、講義なしでゼロから学ばせるのは時間的にも困難があるので（課題研究は「講義はせず、生徒に自主的な研究をさせる」という方針で行なっている）、Web上のCinderellaJapanのサイトにレクチャを書いた。生徒はそれを読んで自分でコーディングを行なった。CinderellaJapanのサイトでは、「3Dグラフィクスのための基本パッケージ」を提供しているが、下図はこれを利用したのではなく、生徒がレクチャを読んで独自に作成したものである。



ローレンツアトラクタの3次元相空間描画


3.3 初期値鋭敏性を示すアニメーション

コンピュータの内部時計を利用し、カオスの特徴である初期値鋭敏性を如実に示す作品を作った。 t を時刻とし、内部時計を取得してアニメーションにする。わずかに初期値が異なる2点が、しだいに別の軌道を描くようになり、十数秒後にはまったく異なる軌道に移る。ローレンツアトラクタでの初期値鋭敏性の意味がよくわかる作品であり、調べた限りでは書籍やWeb上には同様のものは見当たらず、生徒が考えて作成したものである。これにより、カオスの意味についての理解が一層進んだものと思われる。




3.4 独自のストレンジアトラクタ

物理現象を数式化し、そこからカオスを発見するのは高校生の課題研究では無理である。しかし、どのような微分方程式からカオスが現れるかを試行錯誤的に発見するのであれば可能性がある。結果として、ローレンツモデルとダフィン方程式に類似した方程式からいくつかのストレンジアトラクタを描くことができた。ただし、これが、力学系としてのカオスの条件（リャプノフ指数など）を満たすかどうかといった理論的検証は高校生の範疇を超えているので、相空間内における図形的特質からカオスであろう、ということにとどまっている。



ローレンツ型



ジャパニーズ型

4 ストレンジアトラクタを描画する

ここからは、CindyScriptによるストレンジアトラクタの描画と、その関連事項について例を示し、数学教育におけるCindyScriptの有用性を論じる。

カオスの例としてよく挙げられるのが、ローレンツアトラクタとジャパニーズアトラクタである。いずれも非線形微分方程式において、係数がある範囲の値をとるときにス

トレンジアトラクタが現れる。ただし、ジャパニーズアトラクタはポアンカレ断面を表示するのが一般的である。これらは、カオスの入門書や啓蒙書に必ずといっていいほど登場するが、きちんと調べていくと奥深いものがある。

4.1 ローレンツアトラクタ

ローレンツアトラクタは、ローレンツ方程式と呼ばれる次の微分方程式において、係数がある値をとるときに現れる。

$$\frac{dx}{dt} = -\sigma x + \sigma y, \quad \frac{dy}{dt} = -xz + rx - y, \quad \frac{dz}{dt} = xy - bz$$

この非線形微分方程式の解を相空間にプロットするために、オイラー法またはルンゲクッタ法を用いる。オイラー法は、微分の定義式から簡単に導かれるため、高校生にもわかりやすく、コーディングも簡単である。次図にコードと結果を示す。

```
s=10;
b=8/3;
r=28;
dt=0.004;

fx(x,y):=x-dt*s*(x-y);
fy(x,y,z):=y+dt*(-x*z+r*x-y);
fz(x,y,z):=z+dt*(x*y-b*z);

work=[10,12,25]; //初期値

repeat(4000,
  x=work_1;
  y=work_2;
  z=work_3;
  work=[fx(x,y),fy(x,y,z),fz(x,y,z)];

  listxy=[fx(x,y),fy(x,y,z)];
  listxz=[fx(x,y),fz(x,y,z)];

  draw(listxy,size->1,color->[1,0,0],noborder->true);
  draw(listxz,size->1,color->[0,0,1],noborder->true);
);
```

ローレンツアトラクタの描画とスクリプト

係数 s , b , r の値と微小区間の値 dt を設定し、差分方程式を関数として定義する。初期値を `work` というリストで与えておき、`repeat` 関数によって4000回繰り返し処理を行なっている。 xy 平面と xz 平面のものを同一面に描画している。Excel を用いたものなどで行われている方法である。([1]) しかし、相空間は3次元なので、できれば3Dグラフィクスで表示したい。CindyScript を用いた3Dグラフィクスについては、昨年度、Web上にその原理からコーディングまでのレクチャを書いたものがある。スクリプトも与えてあるので、生徒はそれを読んだり、Cinderella のサイト「MatheVital」の作品を解析したりしながら自分でコーディングしていった。3Dの描画は、CindyScript の組み込み関数 `draw()` に、引数として `map3D` 関数で3次元の点を2次元に射影した点を与えておこなう。`map3D` 関数はユーザー定義関数で、Initialization スロットに定義されている。したがって、差分方程式を用いてプロットする中核の部分だけ微分方程式に応じて書き換えれば、3Dのストレンジアトラクタを描くことができる。

```

// 角θの回転行列
rtx(θ):=[[1,0,0],[0,cos(θ),-sin(θ)],[0,sin(θ),cos(θ)]];
rty(θ):=[[cos(θ),0,sin(θ)],[0,1,0],[-sin(θ),0,cos(θ)]];
rtz(θ):=[[cos(θ),-sin(θ),0],[sin(θ),cos(θ),0],[0,0,1]];

// 回転行列初期値
mat=rtx(pi/12)*rty(-2*pi/3)*rtz(-pi/2);

viewpoint=9.001; // 視点のy座標 整数にしないこと
// 視点を遠くすると平行投影に近くなる
zoom=15; // 拡大率

// 透視変換。使い方は
// p=[1,2,3] という3次元座標を
// A.xy=map(p) で2次元座標にしてAの(x,y)座標に返す
map3D(p):=(
  p=mat*p;
  mp=p/(viewpoint-p_3);
  [mp.x,mp.y]*zoom;
);

rotating=true; // 回転フラグ

```

3Dグラフィックスのためのスクリプト

座標軸は、Cinderellaで2点を通る有向直線を3本作図し、これをCindyScriptでコントロールする。次の図で、XX0.xyなどは、Cinderellaで作図した点XX0の座標を示す。

```

// 軸と原点の位置決め
XX0.xy=map3D([-4,0,0]);
XX1.xy=map3D([4,0,0]);
YY0.xy=map3D([0,-4,0]);
YY1.xy=map3D([0,4,0]);
ZZ0.xy=map3D([0,0,-4]);
ZZ1.xy=map3D([0,0,5]);
O.xy=map3D([0,0,0]);

```

Cinderellaの幾何要素を使って座標軸を表示するスクリプト

```

s=10;
b=8/3;
r=28;
dt=0.002;

fx(x,y):=x-dt*s*(x-y);
fy(x,y,z):=y+dt*(-x*z+r*x-y);
fz(x,y,z):=z+dt*(x*y-b*z);


work=[10,12,25]; //初期値

if(Text0.pressed,
  repeat(10000,
    x=work_1;
    y=work_2;
    z=work_3;
    work=[fx(x,y),fy(x,y,z),fz(x,y,z)];
    // 1/10に縮小して表示
    draw(map3D(work/10),size->0.5,color->[1,0,0],
      noborder->true);
  );
);

```

ローレンツアトラクタの3D描画とスクリプト

アトラクタを描画するためのスクリプトはほとんど同じである。さらに、描画面上でマウスをドラッグすれば視点を移動させ、対象を回転することができる。マウスがドラッグされたときの動作は、Mouse Drag スロットに書かれている。ドラッグの方向と量により回転行列を変化させている。



マウスがドラッグされたときの動作

このように、CindyScriptでは、プログラム全体をカテゴリー分けすることにより、構造的なプログラミングができるようになっている。

4.2 ジャパニーズアトラクタ


ジャパニーズアトラクタは、1961年に京都大学の上田が発見した、ダフィン方程式の解曲線のアトラクタで、通常はそのポアンカレ断面を示す。ダフィン方程式は文献によって少しずつ形が異なるが、ここでは上田が大学院生のときに属していた林研究室で研究されていた次の式を扱う。

$$\frac{d^2x}{dt^2} + k\frac{dx}{dt} + x^3 = B \cos t$$

ここで、 $dx/dt = y$ とおくと、1階の連立微分方程式に変形することができる。

$$\frac{dx}{dt} = y \quad , \quad \frac{dy}{dt} = -ky - x^3 + B \cos t$$

この解曲線は、時間 t を含めると3次元の相空間にプロットできるが、 $\cos t$ の周期性を利用して、 $t = 2n\pi$ の値だけを2次元の相空間にプロットしたのがポアンカレ断面である。やはりオイラー法によって差分化すれば、ローレンツアトラクタと同じようなコードで表示することができる。




ジャパニーズアトラクタ

なお、本図では原点付近に楕円状の点の集合があるが、これは初めから180回くらいまでのループで現れるもので、過渡的なものとして通常は破棄する。

4.3 生徒の取り組み

生徒は、以上のローレンツアトラクタとジャパニーズアトラクタを Excel で描画 ([1]) したのち、CindyScript でプログラムを作成した。初めは Web 上のレクチャにあるコードをそのまま使っていたが、原理がわかってくると Cinderella の他のレクチャも参照して、3次元の相空間での描画や、コンピュータの内部時計を用いてのアニメーションを作るようになった。ジャパニーズアトラクタでは、3次元の相空間で解曲線を描画するという、まったく教えていないことにまで踏み込めるようになった。



ジャパニーズアトラクタの3次元解曲線

さらに、微分方程式の項と係数を変化させるアトラクタの探索に乗り出した。その結果、ローレンツ型で2つ、ジャパニーズアトラクタ型で2つのアトラクタを作り出すことができた。その微分方程式は次の通りである。

さらに、微分方程式の項と係数を変化させるアトラクタの探索に乗り出した。その結果、ローレンツ型で2つ、ジャパニーズアトラクタ型で2つのアトラクタを作り出すことができた。その微分方程式は次の通りである。

- ・ローレンツ型その1

$$\frac{dx}{dt} = -a(x - y + z) \quad , \quad \frac{dy}{dt} = b(-xz + cx - y) \quad , \quad \frac{dz}{dt} = xy - bz$$

- ・レスラー型その2 (レスラー型)

$$\frac{dx}{dt} = b - z \quad , \quad \frac{dy}{dt} = x - z \quad , \quad \frac{dz}{dt} = yz + x - r$$

- ・ジャパニーズアトラクタ型その1

$$\frac{dx}{dt} = y - kx^2 \quad , \quad \frac{dy}{dt} = -ky - x^3 + B \cos t + k^2(y - (B + 1 - ky)x^3 - y^2)$$

- ・ジャパニーズアトラクタ型その2

$$\frac{dx}{dt} = y - kx^2 - ky \quad , \quad \frac{dy}{dt} = -ky - x^3 + B \cos t$$

微分方程式としての違いはわずかであるが、係数や項を少し変えただけでストレンジアトラクタは現れなくなることを考えれば、決して簡単に見つかったものではなく、試行錯誤の末に4つだけ見つけたというべきだろう。(図は前掲)

5 結論と課題

以上のように、CindyScript は高校生でも容易に習得でき、プログラミングそのものよりも、これを用いたシミュレーションに多くの労力を割くことができ、探求活動の強力な道具となりうることがわかった。しかし、「ストレンジアトラクタの探究」というテーマに関しては、次の課題がまだ残っている。

- ルンゲクッタ法による数値解

オイラー法よりもルンゲクッタ法の方が誤差の少ない結果が得られる。生徒はルンゲクッタ法にも取り組んではいたが、オイラー法との違いを明確にするなど、論文としてこれを含めるには至らなかった。

- 計算機の誤差問題

大学の講義などでは、コンピュータの小数計算の誤差問題が扱われている。([11]) Excel の計算誤差も有名である。カオスに関しては、ロジスティック写像のリターンマップとそれに関連する計算機の誤差問題がある。これについては、高校生の課題研究の進行と合わせた形で、CinderellaJapan サイトの「Cinderella でカオス」の項目に掲載したが、高校生にとっても格好の研究材料となるだろう。

- ジャパニーズアトラクタの描画に関する問題

前項と関連するが、ジャパニーズアトラクタの描画については微妙な問題がある。文献 [1] では、微小区間 dt として、 $2\pi/800$ をとっている。これを $2\pi/720$ とす

るとアトラクタが現れなくなる。また、 $2\pi/800$ でも、ルンゲクッタ法で計算するとアトラクタが現れない。上田は「複雑系を超えて」([8])の中で、「数値解析のアルゴリズムにしても、どのようなアルゴリズムを使っても、また時間軸の刻みにしても、そこそこ大きくとっていただいても、同じ図がでてきます。」と書いている。しかし、そうはなっていないのである。この点については、生徒も探究をしたが原因は掴めなかった。数値シミュレーションの信頼度に関する、今後の大きな課題である。

これらは、今年の研究を先行研究として、後の学年の生徒が取り組むことを期待している。

参考文献

- [1] 白田・伊藤・井上,Excelで学ぶ理工系シミュレーション入門,(CQ出版社),2003
- [2] E.N.Lorenz, Deterministic Nonperiodic Flow , Journal of the ATMOSPHERIC SCIENCES ,1963
- [3] ジェイムズ・グリック,カオス 新しい科学をつくる,新潮文庫,1991
- [4] E.N. ローレンツ,カオスのエッセンス,共立出版,1997
- [5] 山口昌哉,カオスとフラクタル,ちくま学芸文庫,2010
- [6] 山口昌哉,カオス入門,朝倉書店,1996
- [7] ラルフ・エイブラハム, ヨシスケ・ウエダ著,稲垣・赤松訳,カオスはこうして発見された,共立出版,2002
- [8] 上田・西村・稲垣,複雑系を超えて カオス発見から未来へ,筑摩書房,1999
- [9] Hirsch , Smale , Devaney, 力学系入門 原著第2版,共立出版,2007
- [10] 林・上田・赤松,ある二階非線形常微分方程式の定常解について,数理解析研究所講究録 113 巻,1971 1-27
- [11] 井上純一,カオス・フラクタル講義ノート,<http://chaosweb.complex.eng.hokudai.ac.jp/>,2009